SEMI-EMPIRICAL NUCLEAR MASS FORMULA

(requires one memory module)

A Semi-Empirical formula is used to calculate approximate binding energies and mass excess for any nucleus with a given nuclear charge, Z, and number of neutrons, N.

Definition: Binding energy (B.E.) = $Z * M_p + N * M_n - M(Z,N)$

M_p = proton mass (energy) in MeV, M_n = neutron mass in MeV

M(Z,N) = mass of nucleus having Z protons and N neutrons.

Mass Excess = M(Z,N) - A * (amu)

A = Z + N, 1 (amu) = M(6,6)/12 --- 1/12 mass of ^{12}C

Weizsacker's Semi-Empirical mass formula contains seven terms

$$M(Z,N) = Z*M_p + N*M_n + E_v + E_s + E_c + E_{sym} + E_{pair}$$

$E_v = -a1 * A$

$E_s = a2 * A^{2/3}$

$E_c = a3 * Z^2/A^{1/3}$

$E_{sym} = a4 * (z-N)^2/A$

$E_{pair} = \pm 34/A^{3/4}$ depending on whether Z and N are both odd or both even.

$E_{pair} = 0$ for odd A nuclei

Notes:

The semiempirical formula has been derived from measured masses and binding energies and is expected to work for nuclei reasonably close to the valley of stability. Usually $N \geq Z$ especially for heavier nuclei.

Example:

What is the binding energy, the mass, mass excess, volume energy, surface energy, coulomb energy, symmetry energy, and pairing energy of the element titanium (Z = 22, N = 26)?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 025	
[XEQ] [ALPHA] NM [ALPHA]	NUM PROT?
22 [R/S]	NUM NEUT?
26 [R/S]	B.E.=-404.5143
[R/S]	B.E./A=-8.4274
[R/S]	M=44,677.9077
[R/S]	M/A=930.7897
[R/S]	M.E.=0.0000
[R/S]	M.E./A=0.0000
[R/S]	EV=-752.6400
[R/S]	EV/A=-15.6800
[R/S]	ES=245.1351
[R/S]	ES/A=5.1070
[R/S]	Ec=95.4884
[R/S]	Ec/A=1.9893
[R/S]	ESYM=9.3667
[R/S]	ESYM/A=0.1951
[R/S]	EP=-1.8644
[R/S]	EP/A=-0.0388

User Instructions

SIZE: 025

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1.	Load program.			
2.	Initialize program.		[XEQ] NM	NUM PROT?
3.	Input number of protons.	Z	[R/S]	NUM NEUT?
4.	Input number of neutrons	N	[R/S]	B.E.=
	and solve for binding energy terms.		[R/S]	B.E./A=
5.	Calculate the mass of nucleus.		[R/S]	M=
5a.	Calculate the mass per nucleon.		[R/S]	M/A=
6.	Calculate the mass excess of nucleus.		[R/S]	M.E.=
6a.	Calculate mass excess per nucleon.		[R/S]	M.E./A=
7.	Calculate the volume energy.		[R/S]	EV=
7a.	Calculate volume energy per nucleon.		[R/S]	EV/A=
8.	Calculate surface energy.		[R/S]	ES=
8a.	Calculate surface energy per nucleon.		[R/S]	ES/A=
9.	Calculate coulomb energy.		[R/S]	Ec=
9a.	Calculate coulomb energy per nucleon.		[R/S]	Ec/A=
10.	Calculate symmetry energy.		[R/S]	ESYM=
10a.	Calculate symmetry energy per nucleon.		[R/S]	ESYM/A=
11.	Calculate pairing energy.		[R/S]	EP=
11a.	Calculate pairing energy per nucleon.		[R/S]	EP/A=
12.	Reset for new case.		[R/S]	NUM PROT?

Program Listings

```
01*LBL "NM"
02 CF 01
03 CF 00
04 CLRG
05 -931.504
06 STO 08
07 938.793
08 STO 09
09 939.576
10 STO 10
11 -15.68
12 STO 11
13 18.56
14 STO 12
15 .717
16 STO 13
17 28.1
18 STO 14
19 -17
20 STO 15
21 "NUM PRO
T?"
22 PROMPT
23 FS?C 22
24 STO 01
25 "NUM NEU
T?"
26 PROMPT
27 FS?C 22
28 STO 02
29 RCL 02
30 RCL 01
31 +
32 STO 00
33 XEQ C
34 XEQ D
35 XEQ E
36 XEQ a
37 XEQ b
38 XEQ c
39 XEQ d
40 XEQ e
41 GTO "NM"
42*LBL C
43 XEQ 01
44 19.024
45 STO 03
46 XEQ 04
47 "B.E.="
48 ARCL X
49 PROMPT
```

Initialize and store constants

Prompt for inputs

Calculate Binding Energy

```
50 RCL 00
51 /
52 "B.E./A=
"
53 ARCL X
54 PROMPT
55 RTN
56*LBL D
57 17.024
58 STO 03
59 XEQ 04
60 "M="
61 ARCL X
62 PROMPT
63 RCL 00
64 /
65 "M/A="
66 ARCL X
67 PROMPT
68 RTN
69*LBL E
70 XEQ 01
71 16.024
72 XEQ 04
73 "M.E.="
74 ARCL X
75 PROMPT
76 RCL 00
77 /
78 "M.E./A=
"
79 ARCL X
80 PROMPT
81 RTN
82*LBL a
83 RCL 20
84 "EV="
85 ARCL X
86 PROMPT
87 RCL 00
88 /
89 "EV/A="
90 ARCL X
91 PROMPT
92 RTN
93*LBL b
94 RCL 21
95 "ES="
96 ARCL X
97 PROMPT
98 RCL 00
```

Calculate mass

Calculate mass excess

Display volume energy

Display surface energy

Program Listings

```
 99 /
100 "ES/A="
101 ARCL X
102 PROMPT
103 RTN
104*LBL c
105 RCL 22
106 "Ec="
107 ARCL X
108 PROMPT
109 RCL 00
110 /
111 "Ec/A="
112 ARCL X
113 PROMPT
114 RTN
115*LBL d
116 RCL 23
117 "ESYM="
118 ARCL X
119 PROMPT
120 RCL 00
121 /
122 "ESYM/A=
"
123 ARCL X
124 PROMPT
125 RTN
126*LBL e
127 RCL 24
128 "EP="
129 ARCL X
130 PROMPT
131 RCL 00
132 /
133 "EP/A="
134 ARCL X
135 PROMPT
136 RTN
137*LBL 01
138 8
139 STO 03
140 RCL 00
141 XEQ 02
142 RCL 01
143 XEQ 02
144 RCL 02
145 XEQ 02
146 RCL 00
147 XEQ 02
148 RCL 00
149 3
```

Display Coulomb Energy

Display Symmetry Energy

Display Pairing Energy

Calculation of all terms

```
150 1/X
151 Y↑X
152 X↑2
153 XEQ 02
154 RCL 01
155 X↑2
156 RCL 00
157 3
158 1/X
159 Y↑X
160 /
161 XEQ 02
162 RCL 01
163 RCL 02
164 -
165 X↑2
166 RCL 00
167 /
168 XEQ 02
169 -1
170 RCL 01
171 Y↑X
172 -1
173 RCL 02
174 Y↑X
175 +
176 RCL 00
177 .75
178 Y↑X
179 /
180 XEQ 02
181 SF 01
182 RTN
183*LBL 02
184 RCL IND
03
185 *
186 RCL 03
187 9
188 +
189 STO 03
190 X<>Y
191 STO IND
03
192 8
193 ST- 03
194 RTN
195*LBL 04
196 0
197 STO 06
198*LBL 05
199 ISG 03
```

Program Listings

```
200 GTO 10
201 GTO 06
202*LBL 10
203 RCL IND
03
204 ST+ 06
205 GTO 05
206*LBL 06
207 RCL 06
208 RTN
209 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	A	50	
	Z		
	N		
	indirect adress		
05		55	
	used		
	-amu		
	Mp		
10	Mn	60	
	$-a_1$		
	a_2		
	a_3		
	a_4		
15	$-a_5$	65	
	Z Mp		
	N Mn		
20	EV	70	
	ES		
	Ec		
	ESYM		
	EP		
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 025 TOT. REG. 89 USER MODE
ENG ____ FIX 4 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Used	
01		Used	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY